Illustration of the Effects of Measurement Error: Regression and Mediation Models

Effects of Measurement Error on Regression Analysis

Measurement error biases the estimates of bivariate relationships (e.g., correlation, simple regression) by reducing the magnitude (attenuation) of the relationship, leading to the greater likelihood that a true relationship may be missed. As described in the handout "Latent Variables," attenuation occurs because observed measures have higher variance than the true score of the measure to the extent that measurement error is present in the observed measure. For multiple regression, the effects of measurement error are more complex. Consider a multiple regression with two predictors.



The equation below is for the standardized (partial) regression coefficient (β_2^*) for a model with two predictors, x_1 and x_2, where r_{y2} represents the correlation between x_2 and y and r_{y1} represents the correlation between x_1 and y, and r_{12} represents the correlation between x_1 and x_2.

$$\beta_2^* = \frac{r_{y2} - r_{y1}r_{12}}{1 - r_{12}^2}$$

One can think about this equation as showing that the partial regression coefficient, β_2^* , is a correction to the bivariate correlation r_{y2}, in which the product of r_{y1} and r_{12} are subtracted. Where either of the correlations involved in the correction term (i.e., the correlations between x_1 and y or between x_1 and x_2) are underestimated because of measurement error, there will be less of a correction and the partial regression coefficient will tend to be too large. The effect is complex, however, because the original correlation between the predictor x_2 and y, r_{12}, will tend toward underestimation due to measurement error; whereas the partial regression coefficient will tend toward being overestimation because measurement error in the covariate and the outcome. These directions of bias, however, assumes a more typical effect of the correlations on the partial regression coefficient. But, in the case of suppression, for example, the effects of measurement error might work differently. Thus, there is a somewhat unpredictable effect of measurement error on partial regression coefficients, leading to a degree of inaccuracy in the coefficient estimate that will be larger to the extent that there is more measurement error and more variables involved. The total variance accounted for in the dependent variable as measured by R^2 will be underestimated to the extent that any of the variables contain measurement error. The standard errors will also tend to be inflated due to the additional variance when there is measurement error.

The impact of measurement error on mediation analysis is similarly complex (Cole & Preacher, 2014; Fritz et al., 2016). Simply put, traditional mediation analysis with measured variables assumes no measurement error in the predictors; and, in terms of standardized effects, total variance accounted for, and significance, it is assumed that there is no measurement error in any of the predictors. For a model specified as a full mediational model with no direct c' effect, both the a and the b paths are simple regression coefficients (assuming no covariates). But when the direct effect is included, as is typical when the indirect effect coefficient is computed, the b path in the mediational model is a partial regression coefficient because the predictor, x, is controlled.



The direction and the degree of the bias for estimating the indirect effect will depend on the extent to which each of the variables is measured with measurement error (Hoyle & Kenny, 1999).

Multiple Regression Path Model Example with Measured Variables

The following examples compare a multiple regression model using measured vs. latent variables and a mediational model using measured vs. latent variables. The most important difference between the results with the measured and the latent variables is that the latent variable models remove measurement error (or unique variance more precisely), whereas the measured variable models do not. Another difference, which is likely to have a subtler impact (Fava & Velicer, 1992) is that the latent variable unequally weights the items, whereas the composite scale scores used equal weighting of the items when I computed them.

Below I illustrate a regression path model in Mplus using data from a study of caregivers and care recipients (Newsom, 1999). The variables used in the analysis are composite indexes (using a mean or sum of the items) of a 4-item marital difficulty scale, a 7-item self-esteem scale (Rosenberg self-esteem), and a 9-item depression scale (CES-D 9). For brevity, I only show the standardized output.[1] Please note that I use standard maximum likelihood estimation with no missing data for simplicity and to have greater consistency with regression analysis, but later we will include robust standard errors and include missing data.

```
title:  Example of effects of measurement error on regression and mediation;
data:  file=measerr.dat;  format=free;

variable:  names = mardiff selfest dep
    rsclose rsinvol rstakes rsdays
     rnotworr rnumqal ramfailr ramable rnotprdr rfelpos rfelsat
      rbother rblues rkpmind rdepres reffort rrestls rwashap rsad renjoylr;
    !there are no missing data in this data set;

    usevariables=mardiff selfest dep

analysis:  type=general;  estimator=ml;

model:  dep on mardiff selfest;
          mardiff with selfest;

output:  stdyx;

STANDARDIZED MODEL RESULTS
STDYX Standardization
                                                       Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value

 DEP      ON
    MARDIFF            0.243      0.091      2.685      0.007
    SELFEST           -0.299      0.089     -3.342      0.001

 MARDIFF  WITH
    SELFEST           -0.458      0.073     -6.251      0.000
```

Multiple Regression Model Example with Latent Variables

Below are the results from the same model using latent variables. Again, only the standardized output is shown.

```
title:  Example of effects of measurement error on regression and mediation;
data:  file=measerr.dat;  format=free;

variable:  names = mardiff selfest dep
  rsclose rsinvol rstakes rsdays
   rnotworr rnumqal ramfailr ramable rnotprdr rfelpos rfelsat
    rbother rblues rkpmind rdepres reffort rrestls rwashap rsad renjoylr;
  !there are no missing data in this data set;

 usevariables=  rsclose rsinvol rstakes rsdays
   rnotworr rnumqal ramfailr ramable rnotprdr rfelpos rfelsat
    rbother rblues rkpmind rdepres reffort rrestls rwashap rsad renjoylr;
```

[1] Another approach to demonstrating the effect of measurement error is to specify a measurement error correction directly by creating a latent variable with just one indicator, setting the loading equal to 1 and the measurement residual variance (measurement error) equal to (1-reliability) × (variance of the variable). Although this approach to path models may offer some advantages compared with regression analysis, which assumes no measurement error in the variables, I do not recommend this approach except for in relatively desperate circumstances. Deshon (1998) and MacDonald (1996) provide good discussions of some of the issues and potential pitfalls involved in this approach to error correction in path models.

```
analysis:  type=general;  estimator=ml;

model:
 mardiff by rsclose-rsdays;
 selfest by rnotworr-rfelsat;
 dep by rbother-renjoylr;

 dep on mardiff selfest;
      mardiff with selfest;

output:  stdyx;
STANDARDIZED MODEL RESULTS
STDYX Standardization

 DEP      ON
    MARDIFF            0.099      0.145      0.684      0.494
    SELFEST           -0.539      0.126     -4.267      0.000

 MARDIFF  WITH
    SELFEST           -0.571      0.102     -5.599      0.000
```

Summary Table Comparing the Regression Results with and without Latent Variables

	Standardized Coefficients	
	Path Model	**Latent SEM Model**
Marital Difficulty Predicting Depression, β_1^*	.243**	.099
Self-esteem Predicting Depression, β_2^*	-.299**	-.539***
Marital Difficulty-Self-Esteem Correlation, r_{12}	-.458***	-.571***

Mediational Path Model Example with Measured Variables

Marital Difficulty → *Self-esteem* → *Depression*

```
STANDARDIZED MODEL RESULTS
STDYX Standardization
                                                        Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value

 DEP      ON
    MARDIFF            0.243      0.077      3.169      0.002
    SELFEST           -0.299      0.091     -3.291      0.001

 SELFEST  ON
    MARDIFF           -0.458      0.073     -6.278      0.000


STANDARDIZED TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS
STDYX Standardization
                                                        Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value

Effects from MARDIFF to DEP

  Total                0.380      0.064      5.959      0.000
  Total indirect       0.137      0.051      2.661      0.008

CONFIDENCE INTERVALS OF TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS


                  Lower .5%  Lower 2.5%    Lower 5%    Estimate    Upper 5%  Upper 2.5%   Upper .5%

Effects from MARDIFF to DEP

  Total              2.046       2.321       2.538       3.553       4.637       4.924       5.370
  Total indirect     0.214       0.446       0.559       1.280       2.217       2.493       2.915
```

Mediational Model Example with Latent Variables

```
analysis:  type=general;  estimator=ml;  matrix=covariance;
    bootstrap = 1000;
```

```
model:
 mardiff by rsclose-rsdays;
 selfest by rnotworr-rfelsat;
 dep by rbother-renjoylr;

 dep on mardiff selfest;
  selfest on mardiff;

 Model indirect:  dep ind mardiff;

output: stdyx cinterval(bootstrap);

STANDARDIZED MODEL RESULTS
STDYX Standardization

DEP      ON
    MARDIFF           0.099      0.268      0.371      0.711
    SELFEST          -0.539      0.288     -1.871      0.061

 SELFEST  ON
    MARDIFF          -0.571      0.133     -4.279      0.000

STANDARDIZED TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS
                                                         Two-Tailed
                    Estimate       S.E.  Est./S.E.    P-Value

Effects from MARDIFF to DEP

  Total               0.407      0.093      4.393      0.000
  Total indirect      0.308      0.261      1.178      0.239

CONFIDENCE INTERVALS OF TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS


                  Lower .5%  Lower 2.5%    Lower 5%    Estimate    Upper 5%  Upper 2.5%   Upper .5%

Effects from MARDIFF to DEP

  Total             0.031       0.066       0.097       0.287       0.544       0.654       0.791
  Total indirect   -0.011       0.028       0.043       0.217       0.515       0.583       1.436
```

Summary Table Comparing the Mediation Results with and without Latent Variables

	Standardized Coefficients	
	Path Model	**Latent SEM Model**
Marital Difficulty Predicting Self-Esteem (*a*)	-.458***	-.571***
Self-esteem Predicting Depression (*b*)	-.299**	-.539***
Marital Difficulty Predicting Depression (c')	.243**	.099
Indirect Effect Coefficient (*ab*)	.137*	.308*

Note the significance of the indirect effect I indicated in the table is based on the percentile bootstrap confidence intervals. The path model supports partial mediation, whereas the latent variable model supports full mediation. As with any mediational model, there will be a complex consequence of measurement error for the pattern of correlations, the difference in the strength of the direct effects in the two models, and the impact that the higher correlations of x and m with y have on the standard error.

References and Further Reading

Cole, D. A., & Preacher, K. J. (2014). Manifest variable path analysis: potentially serious and misleading consequences due to uncorrected measurement error. *Psychological Methods, 19*(2), 300-315.
Deshon, R.P. (1998). A cautionary note on measurement error corrections in structural equation models. *Psychological Methods, 3*, 412-423.
Fava, J.L., & Velicer, W. F. (1992). An empirical comparison of factor, image, component, and scale scores. *Multivariate Behavioral Research, 27*, 301-322.
Fritz, M. S., Kenny, D. A., & MacKinnon, D. P. (2016). The combined effects of measurement error and omitting confounders in the single-mediator model. *Multivariate behavioral research, 5*1, 681-697.
Hoyle, R.H. & Kenny, D.A. (1999). Sample size, reliability, and tests of statistical mediation. In: Hoyle, RH., editor. *Statistical strategies for small sample research* (pp.195-222). Thousand Oaks, CA: Sage.
MacKinnon, D. P. (2008). Chapter , *Introduction to Statistical Mediation Analysis*. New York: Erlbaum.
McDonald, R.P. (1996). Path analysis with composite variables. *Multivariate Behavioral Research, 31*, 239-270.
Newsom, J.T. (1999). Another side to caregiving: Negative reactions to being helped. *Current Directions in Psychological Science*, *8*, 183-187.